

Representación gráfica de curvas

Los objetivos de esta práctica son:


- Representar gráficamente curvas planas descritas en forma paramétrica, implícita o polar.

1. Curvas planas

Curvas dadas en forma explícita

Entendemos por esto la gráfica de una función real de una variable. Aunque no vamos a precisar más, se suele imponer alguna condición a la función para llamar curva a su gráfica; por ejemplo, que sea continua, o que sea derivable o derivable a trozos o derivable hasta cierto orden prefijado. En las prácticas anteriores ya hemos visto cómo se representa una gráfica con las órdenes **plot** y **display**, y conocemos diversas opciones que nos permiten afinar la presentación, como **color**, **thickness**, **discont**, **scaling**,... Algunas de estas órdenes, en particular **display**, necesitan cargar antes el paquete **plots**. Veamos un ejemplo:


```
> with(plots): f:=x->(x^2+2)/(x-3);
      f:=x-> $\frac{x^2+2}{x-3}$ 
> curva:=plot(f(x), x=-15..25, y=-10..30, color=red, thickness=3,
  discont=true):
  asint1:=plot(x+3, x=-15..-5, color=blue, thickness=2):
  asint2:=plot(x+3, x=5..25, color=black, thickness=2):
> display(curva, asint1, asint2, scaling=constrained);
```



Curvas en forma paramétrica


Una curva puede darse también en forma paramétrica, es decir, describiendo los puntos (x,y) de la curva mediante dos funciones: $x=f(t), y=g(t)$. La variable t se suele llamar el parámetro de la curva. Por ejemplo, la circunferencia centrada en el origen y radio 1 se puede describir mediante $x=\cos(t), y=\sin(t)$, donde $t \in [0, 2\pi]$. En Maple pueden representarse estas curvas:

```
> restart; plot([cos(t), sin(t), t=0..2*Pi]);
```




Parece una elipse. Eso es porque la escala no es la misma en los dos ejes y la figura aparece distorsionada. Podemos pedir que la escala sea la misma con la orden **scaling=constrained**.

```
> plot([cos(t), sin(t), t=0..2*Pi], scaling=constrained,
  color=blue, thickness=2);
```




Cualquier curva dada en forma explícita $y=f(x)$, también se puede escribir trivialmente en forma paramétrica, como $x=t, y=f(t)$. Veamos la curva $y=\frac{x^2+2}{x-3}$ del apartado anterior.

```
> plot([t, (t^2+2)/(t-3), t=-15..25], view=[-15..25, -10..30],
  scaling=constrained, discont=true);
```



Puede representarse más de una curva a la vez. El parámetro no tiene por qué llamarse t . En este ejemplo se llama u en ambas gráficas, pero puede tener cualquier otro nombre y tener diferente nombre en cada curva. Y los parámetros pueden recorrer intervalos distintos en cada curva.


```
> plot([cos(u), sin(u), u=0..2*Pi], [2*cos(u), sin(u), u=-Pi..Pi],
      scaling=constrained, color=[red, blue]);
```



También se puede usar la orden **display**. Seguramente es recomendable usarla en este caso, para que las órdenes queden más claras. Para usar esta orden hay que cargar el paquete **plots**.


```
> circunferencia:=plot([cos(u), sin(u), u=0..2*Pi], color=red);
  ellipse:=plot([2*cos(t), sin(t), t=0..2*Pi], color=blue);
  recta:=plot((x+1)/2, x=-2..2, color=black);
```

```
> with(plots):display(circunferencia, ellipse, recta);
```



Podemos añadir opciones para cambiar el aspecto de la gráfica. Por ejemplo:


```
> display(circunferencia, ellipse, recta, view=[-2..2, -1..1],
      scaling=constrained, tickmarks=[5, 3]);
```



Curvas en forma implícita


Otra forma de indicar una curva plana es como las soluciones de una ecuación en dos variables: $f(x, y) = 0$. Por ejemplo, la circunferencia centrada en el origen y de radio 1 se puede indicar como $x^2 + y^2 = 1$. Esta manera de describir una curva se llama en forma implícita. Y aunque no podemos entrar aquí en detalles, hay que destacar que no todas las ecuaciones de este tipo definen lo que podemos entender por curva. En Maple se pueden representar los puntos (x, y) que cumplen la ecuación $f(x, y) = 0$, mediante la orden **implicitplot**, que pertenece al paquete **plots**. Veamos un ejemplo:

```
> restart:with(plots):
> implicitplot(x^2+y^2=1, x=-2..2, y=-2..2);
```




También se pueden añadir la mayoría de las opciones que ya conocemos.

```
> implicitplot(x^2+y^2=1, x=-2..2, y=-2..2, scaling=constrained,
  color=blue, thickness=3);
```



Otro ejemplo, en el que mezclamos una curva en forma explícita, otra en forma paramétrica y una tercera en forma implícita. Aquí es inevitable usar la orden **display**:


```
> explicit:=plot(1-x^2, x=-2..2, color=red, thickness=2):
parametric:=plot([cos(t), sin(t), t=0..2*Pi], color=blue,
thickness=3):
implicit:=implicitplot(x^2/4+y^2=1, x=-2..2, y=-2..3,
color=black, thickness=2):
> display(explicit, parametric, implicit, scaling=constrained,
view=[-2..2, -1..1], tickmarks=[5, 3]);
```



Curvas en forma polar


Otra forma habitual de describir una curva es polares: no se indican las coordenadas (x, y) de los puntos, sino su módulo ρ y un argumento θ . x, y, ρ, θ están relacionadas mediante $x = \rho \cos \theta$, $y = \rho \sin \theta$. O, en sentido contrario, $\rho = \sqrt{x^2 + y^2}$, y fórmulas adecuadas para θ . Por ejemplo, la curva polar $\rho = 1 - \cos \theta$ equivale a la forma implícita $\sqrt{x^2 + y^2} = 1 - \frac{x}{\sqrt{x^2 + y^2}}$.

```
> restart:with(plots):
> implicitplot(sqrt(x^2+y^2)=1-x/sqrt(x^2+y^2), x=-3..3, y=-3..3,
scaling=constrained);
```



Pero la orden **implicitplot** tiene también una opción específica que indica que la ecuación es la forma polar de la curva que se quiere representar. Es la opción **coords=polar**. La gráfica saldrá probablemente más fiel si la hacemos de esta manera.

```
> implicitplot(r=1-cos(theta), r=0..2, theta=0..2*Pi, coords=
polar, scaling=constrained);
```



Otro ejemplo: la espiral $\rho = \theta$. Como curiosidad, añadimos la opción **grid**, que permite fijar la finura con que Maple hace las gráficas. Indica la resolución (en horizontal y en vertical) con que Maple hace la gráfica. Cuidado: cuantos más resolución se pida, más tiempo tarda Maple en calcular la gráfica. El valor por defecto es **grid=[25, 25]**, y en general es suficiente.

```
> implicitplot(r=theta, r=0..40, theta=0..6*Pi+Pi/4, coords=polar,
scaling=constrained, grid=[100, 100]);
```

